CDIFF: A REDUCE PACKAGE FOR COMPUTATIONS IN GEOMETRY OF DIFFERENTIAL EQUATIONS

R. VITOLO

Abstract. We describe CDIFF, a set of symbolic computation packages devoted to computations in the geometry of Differential Equations (DEs, for short). The development was carried out by P. Gragert, P.H.M. Kersten, G. Post and G. Roelofs at the University of Twente, The Netherlands, with latest contributions by R. Vitolo.

The package is distributed on the Geometry of Differential Equations web site http://gdeq.org (GDEQ for short). The 'Twente' part of the package is included in the official REDUCE distribution.

We start from an installation guide for Linux and Windows. Then we focus on concrete usage recipes for the computation of higher symmetries, conservation laws, Hamiltonian and recursion operators for polynomial differential equations. All programs discussed here are shipped together with this manual and can be found at the GDEQ website. The mathematical theory on which computations are based can be found in refs. [12, 13].

1. Introduction

This brief guide refers to using CDIFF, a set of symbolic computation packages devoted to computations in the geometry of DEs. The name of the package, CDIFF, comes from the fact that it is aimed at defining differential operators in total derivatives in order to do computations involving them. Such operators are called *$\mathcal{C}$-differential operators* (see [12]). CDIFF runs in the computer algebra system REDUCE. Recently, REDUCE 3.8 became free software, and can be downloaded here [1]. This was an important motivation for making our computations accessible to a wider public, also through this user guide.

The development of the CDIFF package was started by Gragert and Kersten for symmetry computations in DEs. Then CDIFF was partly rewritten and extended by Roelofs and Post. This part of the CDIFF package consists of 7 files, but only the main three files are documented [8, 9, 10]. This software and the related documentation can be found in the Geometry of Differential Equations (GDEQ for short) web site [2].

Recently, the author of this user guide wrote additional software which runs 'on top' of CDIFF and is especially designed for computations of integrability-related structures (such as Hamiltonian, symplectic and recursion operators) for systemd of differential equations with an arbitrary number of independent or dependent variables. The aim of this manual is to introduce the reader to the above computations.

Date: 2011 July 10.

2000 *Mathematics Subject Classification.* 37K05.

Key words and phrases. REDUCE, Hamiltonian operators, symplectic operators, recursion operators, generalized symmetries, higher symmetries, conservation laws, nonlocal variables.

The readers are warmly invited to send questions, comments, etc., both on the computations and on the technical aspects of installation and configuration of REDUCE, to the author of this document.

Acknowledgements. My warmest thanks are for Paul H.M. Kersten, who explained to me how to use the original CDIFF package for several computations of interest in the Geometry of Differential Equations. I also would like to thank J.S. Krasil'shchik and A.M. Verbovetsky for constant support and stimulating discussions which led me to write this text.

2. Installation

In order to use CDIFF packages you should be able to write REDUCE programs using CDIFF and run them in the REDUCE interactive shell. So, you need two programs: a recent version of REDUCE with CDIFF package and a text editor preferably oriented to program development.

We stress that *in Windows* most of the technical difficulties related to installation and configuration are due to the lack of a REDUCE installer.

2.1. **Installation of REDUCE with CDIFF.**

In order to install REDUCE it is enough to download from here [1] a precompiled binary for your operating system (*e.g.*, 32-bit or 64-bit Debian-based Linux like Debian itself or Ubuntu, 32-bit Windows) and uncompress it in your computer in a location of your choice. Precompiled binaries starting from 8 October 2010 contain CDIFF in the form of a REDUCE package.

In Linux you can also download `.deb` packages at the GetDeb website [3].

For the moment CDIFF has been tested under Linux (both 32bit and 64bit) and Windows XP; please contact the author of this guide if you tested the package with positive results under Mac or other versions of Windows like Vista or Windows 7.

From now on we will assume that the binary executable of REDUCE is in the path of the executables of your operating system. A typical location in Linux would be `/usr/local/bin`. You might put a link instead of the binary executable.

A REDUCE program using CDIFF package can be written with any text editor; it is customary to use the extension `.red` for REDUCE programs, like `program.red`. If you wish to run your program, just run the REDUCE executable. After starting REDUCE, you would see something like

```
Reduce (Free CSL version), 08-Oct-10 ...

1:
```

At the prompt `1:` write `in "program.red";`. Of course, if the program file `program.red` *is not* in the place where the REDUCE executable is, you should indicate the full path of the program, and this depends on your system. In Linux, assuming that you are the user `user` and your program is in the subdirectory `Reduce/computations` of your home directory, you have something like

```
in "/home/user/Reduce/computations/program.red";
```

In Windows, assuming that you are the user `user` and your program is in the subdirectory `Reduce\computations` of the Desktop folder, you would write

```
  in "C:\Documents and Settings\user\Desktop\Reduce
\computations\program.red";
```

Remember that each time you run REDUCE from a command shell, REDUCE inherits your current path from the shell unless you use an absolute path as above. However, if you start REDUCE with the graphical interface (see below) you can always use the leftmost menu item `File>Open...` in order to avoid to write down the whole absolute path.

2.2. **Choice of an editor for writing REDUCE programs.** Now, let us deal with the problem of writing REDUCE programs.

Generally speaking, any text editor can be used to write a REDUCE program. A more suitable choice is an editor for programming languages. Such editors exist in Linux and Windows, a list can be found here [5].

A suggested text editor in Windows is `notepad++`. This editor is easy to install, it has support for many programming languages (but *not* for REDUCE!), and has a GPL free license, see [4]. Similar tools in Linux are `kwrite` and `gedit`.

However, the only IDE (Integrated Development Environment) for developing programs and running them inside the editor itself exists for the great text editor `emacs`, which runs in all operating systems, and in particular Linux and Windows. We stress that an IDE makes the developing-running-debugging cycle much faster because every step is performed in the same environment.

Installation of `emacs` in Linux is quite smooth, although it depends on the Linux distribution; usually it is enough to select the package `emacs` in your favourite package management tool, like `aptitude, synaptic,` or `kpackage`. In order to install `emacs` on Windows one has to work a little bit more. See here [6] for more information. Assuming that `emacs` it is installed and working, the REDUCE IDE for `emacs` can be found here [11]. We refer to their guide for the installation (the procedure is the same for both Linux and Windows). I tested the IDE with emacs 23.2.1 under Debian-based Linux systems (Debian Etch and Squeeze 32-bit and 64-bit, Ubuntu 11.04 64-bit) and Windows XP and it works fine for me.

Suppose you have `emacs` and its REDUCE IDE installed, then there is a last configuration step that will make `emacs` and REDUCE work together. Namely, when opening for the first time a REDUCE program file with `emacs`, go to the `REDUCE>Customize...` menu item and locate the 'REDUCE run Program' item. This item contains the command which is issued by `emacs` from the REDUCE IDE when the menu item `Run REDUCE>Run REDUCE` is selected. Change the command to:

- under Linux (user and location as above):

  ```
  reduce -w
  ```

- under Windows (user and locations as above):

  ```
  reduce.exe
  ```

This setting will run REDUCE inside `emacs`. If you prefer the (slower) graphical interface to REDUCE, remove '`-w`'. Note that the graphical interface will produce LaTeX output, making it much more readable. This behaviour can be turned off in the graphical interface by issuing the command `off fancy;`.

3. Working with CDIFF

All programs that we will discuss in this manual can be found inside the subfolder `test` in the folder which contains this manual. There are some conventions that I adopted on writing programs which use CDIFF.

- Test files have the following names:

  `equationname_typeofcomputation.red`

  where `equationname` stands for the shortened name of the equation (*e.g.* Korteweg–de Vries is always indicated by `kdv`), and `typeofcomputation` stands for the type of geometric object which is computed with the given file, for example symmetries, Hamiltonian operators, etc.. This string also includes a version number. The extension `.red` will tell `emacs` to load the reduce-ide mode (provided you made the installation steps described in the reduce-ide guides).
- More specific information, like the date and more details on the computation done in each version, are included as comment lines at the very beginning of each file.

If you use a generic editor, as soon as you are finished writing a program, you may run it from within REDUCE by following the instructions in the previous section.

In `emacs` with REDUCE IDE it is easier: issuing the command `M-x run-reduce` (or choosing the menu item `Run REDUCE>Run REDUCE`) will split the window in two halves and start REDUCE in the bottom half. If you are running PSL REDUCE you must first issue the command `lisp set_bndstk_size 1000000;` from within REDUCE, in order to avoid memory problems. If you are running CSL REDUCE there is no need of that instruction. Note that CSL and PSL are two different interpreters of Standard Lisp; REDUCE can use only one at a time. The precompiled binaries which are available come with the CSL interpreter. If you wish to try the PSL interpreter then you have to download the source code of REDUCE and recompile it following the instructions on REDUCE website. In any case REDUCE shows up the type of interpreter at startup, see 2.1.

Then you may load the program file that you were editing (suppose that its name is `program.red`) by issuing `in "program.red";` at the REDUCE prompt. In fact, `emacs` lets REDUCE assume as its working directory the directory of the file that you were editing.

Results of a computation consist of the values of one or more unknown. Suppose that the unknown's name is `sym`, and assume that, after a computation, you wish to save the values of `sym`, possibly for future use from within REDUCE. Issue the following REDUCE commands (of course, after you finish your computations!):

```
off nat;
out "file_res.red";
sym:=sym;
shut "file_res.red";
on nat;
```

The above commands will write the content of `sym` into the file `file_res.red`, where `file` stands for a filename which follows the above convention. The command `off nat;` is needed in order to save the variable in a format which could be imported in future REDUCE sessions. If you wish to translate your results in LaTeX, see the package `rlfi` and its own documentation.

Working remotely with REDUCE is not difficult and it is highly recommended for big computations that a server can run more efficiently and without interruptions. A method of choice to do this is described by the following steps:

(1) login to the remote server with `ssh`;
(2) start `emacs` as a daemon on the server by the command `emacs --daemon` (only from version 23.1!);
(3) run `emacsclient -c file.red`. That program will connect to the `emacs` daemon and open the requested file.
(4) run REDUCE (if you installed the reduce IDE everything is easier, otherwise you should open a shell within emacs and issue the command `reduce`);
(5) exit `emacsclient` normally (C-x C-c). This will not kill the daemon, that will keep your computation running until the end.
(6) login again when you wish to check the computation.

In next sections we will describe some examples of computations with CDIFF. The parts which are shared between all examples are described only once. We stress that all computations presented in this document can be downloaded at the GDEQ website [2], and that they are run in the REDUCE environment by typing `in "program.red";` at the REDUCE prompt, as explained above. Moreover, all examples can be run at once by the shell script `cdiff.sh` to test if the system is working properly and results are the same as obtained previously.

Each computation consists of two parts: setting up the jet space and the equation, and solving the problem using suitable ansatz for the unknown functions. We will emphasize this division only in the first example.

Remark. The mathematical framework on which the computations are based can be found in [12].

4. Higher symmetries

In this section we show the computation of (some) higher symmetries of Burgers'equation $B = u_t - u_{xx} + 2uu_x = 0$. The corresponding file is `bur_hsy1.red` and the results of the computation are in `results/bur_hsy1_res.red`.

The idea underlying this computation is that one can use the scale symmetries of Burgers'equation to assign "gradings" to each variable appearing in the equation (in other words, one can use dimensional analisys). As a consequence, one could try different ansatz for symmetries with polynomial generating functions. For example, it is possible to require that they are sum of monomials of given degrees. This ansatz yields a simplification of the equations for symmetries, because it is possible to solve them in a "graded" way, *i.e.*, it is possible to split them into several equations made by the homogeneous components of the equation for symmetries with respect to gradings.

In particular, Burgers'equation translates into the following dimensional equation:

$$[u_t] = [u_{xx}], \quad [u_{xx}] = [2uu_x].$$

By the rules $[u_z] = [u] - [z]$ and $[uv] = [u] + [v]$, and choosing $[x] = -1$, we have $[u] = 1$ and $[t] = -2$. This will be used to generate the list of homogeneous monomials of given grading to be used in the ansatz about the structure of the generating function of the symmetries.

Setting up the jet space and the differential equation. The program that builds total derivatives restricted to the given equation has to be loaded in the beginning:

```
in "cde.red";
```

Then, CDIFF needs to know the variables, their scale degree and the maximal order of derivatives at which it will compute differential consequences of the given equation. The input is done in this way:

```
indep_var:={x,t}$
dep_var:={u}$
odd_var:={p}$
deg_indep_var:={-1,-2}$
deg_dep_var:={1}$
deg_odd_var:={0}$
total_order:=10$
```

Here

- `indep_var` is the list of independent variables;
- `dep_var` is the list of dependent variables;
- `odd_var` is the list of odd variables (not used in this computation – just a dummy variable);
- `deg_indep_var` is the list of scale degrees of the independent variables;
- `deg_dep_var` is the list of scale degrees of the dependent variables;
- `deg_odd_var` is the list of scale degrees of odd variables (not used in this computation);
- `total_order` is the maximal order of derivatives at which the program will compute differential consequences of the given equation;

Two more parameters can be set for convenience:

```
statename:="bur_hsy1_state.red"$
resname:="bur_hsy1_res.red"$
```

These are the name of the output file for recording the internal state of the program `cde.red`, including the total derivatives, and the name of the file containing results of the computation.

We now give the equation in the form of one of the derivatives equated to a right-hand side expression. The left-hand side derivative is called *principal*, and the remaining derivatives are called *parametric*[1]. Parametric coordinates are coordinates on the equation manifold and its differential consequences, and principal coordinates can be deduced from the differential equation and its differential consequences. For scalar evolutionary equations with two independent variables parametric derivatives are of the

[1]This terminology dates back to Riquier, see [18]

type $(u, u_x, u_{xx}, \ldots)$. Note that the system must be in passive orthonomic form; this also means that there will be no nontrivial integrability conditions between parametric derivatives. (Lines beginning with % are comments for REDUCE.)

```
% left-hand side of the differential equation
principal_der:={u_x0t1}$
% right-hand side of the differential equation
de:={u_x2t0+2*u_x0t0*u_x1t0}$
% same construction for odd coordinates
principal_odd:={p_x0t1}$
de_odd:={-p_x2t0+2*u_x0t0*p_x1t0}$
```

In this computation the odd equation will not have any role, but it must be present even for purely even computations. In order to speed up computations one could set `de_odd` to be zero.

The main routine in `cde.red` is called as follows:

```
cde({indep_var,dep_var,odd_var,total_order},
   {{principal_der,de},{principal_odd,de_odd}})$
```

The function `cde` defines total derivatives truncated at the order `total_order` and restricted on the (even and odd) equation; this means that total derivatives are tangent to the equation manifold. Their coordinate expressions are of the form

$$D_\lambda = \frac{\partial}{\partial x^\lambda} + \sum_{u^i_{\boldsymbol{\sigma}} \text{ parametric}} u^i_{\boldsymbol{\sigma}\lambda} \frac{\partial}{\partial u^i_{\boldsymbol{\sigma}}} + \sum_{p^i_{\boldsymbol{\sigma}} \text{ parametric}} p^i_{\boldsymbol{\sigma}\lambda} \frac{\partial}{\partial p^i_{\boldsymbol{\sigma}}}, \tag{1}$$

where $\boldsymbol{\sigma}$ is a multiindex. It can happen that $u^i_{\boldsymbol{\sigma}\lambda}$ (or $p^i_{\boldsymbol{\sigma}\lambda}$) is principal and must be replaced with differential consequences of the equation. Such differential consequences are called *primary differential consequences*, and are computed; in general they will depend on other, possibly new, differential consequences, and so on. Such newly appearing differential consequences are called *secondary differential consequences*. If the equation is in passive orthonomic form, the system of all differential consequences (up to the maximal order `total_order`) must be solvable in terms of parametric derivatives only. The function *cde* automatically computes all necessary and sufficient differential consequences which are needed to solve the system.

Note that when in total derivatives there is a coefficient of order higher than maximal this is replaced by the string `letop`. If such a string appears during computations it is likely that we went too close to the highest order variables that we defined in the file. This could mean that we need to extend the operators and variable list, just by increasing the number `total_order`. Later on we will describe a useful test to check the absence of `letop` from a computation.

The output generated by the function `cde` is not a result of the computation, but it can be useful for debugging purposes or for storing intermediate computations to be reused later. It can be saved by the function:

```
save_cde_state(statename)$
```

Defining and solving the problem. Higher symmetries of the given equation are functions `sym` depending on parametric coordinates up to some jet space order. We assume that they are graded polynomials of all parametric derivatives. In practice, we generate a linear combination of graded monomials with arbitrary coefficients, then we plug it in the equation of the problem and find conditions on the coefficients that fulfill the equation. To construct a good ansatz, it is required to make several attempts with different gradings, possibly including independent variables, etc.. For this reason, ansatz-constructing functions are especially verbose. In order to use such functions they must be initialized with the following command:

```
cde_grading(deg_indep_var,deg_dep_var,deg_odd_var)$
```

We need one operator `equ` whose components will be the equation of higher symmetries and its consequences. Moreover, we need an operator `c` which will play the role of a vector of constants, indexed by a counter `ctel`:

```
ctel:=0;
operator c,equ;
```

We prepare a list of variables ordered by scale degree:

```
graadlijst:=der_deg_ordering(0,all_parametric_der)$
```

The function `der_deg_ordering` is defined in `cde.red`. It produces the given list using the list `all_parametric_der` of all parametric derivatives of the given equation up to the order `total_order`. The first two parameters can assume the values 0 or 1 and say that we are considering even variables and that the variables are of parametric type.

Then, due to the fact that *all parametric variables have positive scale degree* then we prepare the list `ansatz` of all graded monomials of scale degree from 0 to 5

```
graadmon:=for i:=1:5 collect mkvarlist1(i,i)$
graadmon:={1} . graadmon$
ansatz:=for each el in graadmon join el$
```

More precisely, the command `mkvarlist1(i,i)` produces a list of monomials of degree `i` from the list of graded variables `graadlijst`; the second command adds the zero-degree monomial; and the last command produces a single list of all monomials.

Finally, we assume that the higher symmetry is a graded polynomial obtained from the above monomials (so, it is independent of x and t!)

```
sym:=(for each el in ansatz sum (c(ctel:=ctel+1)*el))$
```

Next, we define the equation $\bar{\ell}_B(\texttt{sym}) = 0$, where $B = 0$ is Burgers'equation and `sym` is the higher symmetry:

```
equ 1:=ddt(sym)-ddx(ddx(sym))-2*u_x0t0*ddx(sym)-2*u_x1t0*sym ;
```

In the above equation total derivatives with respect to x, t are `ddx`, `ddt`. The list of variables, to be passed to the equation solver:

```
vars:=append(indep_var,all_parametric_der);
```

The number of initial equation(s):

```
tel:=1;
```

Next command initializes the equation solver. It passes

- the equation vector `equ` togeher with its length `tel` (*i.e.*, the total number of equations);

- the list of variables with respect to which the system *must not* split the equations, *i.e.*, variables with respect to which the unknowns are not polynomial. In this case this list is just {};
- the constants'vector `c`, its length `ctel`, and the number of negative indexes if any; just `0` in our example;
- the vector of free functions `f` that may appear in computations. Note that in {`f,0,0` } the second `0` stands for the length of the vector of free functions. In this example there are no free functions, but the command needs the presence of at least a dummy argument, `f` in this case. There is also a last zero which is the negative length of the vector f, just as for constants.

```
initialize_equations(equ,tel,{},{c,ctel,0},{f,0,0});
```

Run the procedure `splitvars` on the first component of `equ` in order to obtain equations on coefficiens of each monomial.

```
splitvars 1;
```

Next command tells the solver the total number of equations obtained after running `splitvars`.

```
pte tel;
```

This command solves the equations for the coefficients. Note that we have to skip the initial equations!

```
for i:=2:te do es i;
;end;
```

One more example file is available; it concerns higher symmetries of the KdV equation. In order to deal with symmetries explicitely depending on x and t it is possible to use REDUCE and CDIFF commands in order to have sym = x*(something of degree 3) + t*(something of degree 5) + (something of degree 2); this yields scale symmetries. Or we could use sym = x*(something of degree 1) + t*(something of degree 3) + (something of degree 0); thiw yields Galilean boosts.

5. Local conservation laws

In this section we will find (some) local conservation laws for the KdV equation $F = u_t - u_{xxx} + uu_x = 0$. Concretely, we have to find non-trivial 1-forms $f = f_x dx + f_t dt$ on $F = 0$ such that $\bar{d}f = 0$ on $F = 0$. "Triviality" of conservation laws is a delicate matter, for which we invite the reader to have a look in [12].

The files containing this example are `kdv_lcl1,kdv_lcl2` and the corresponding results and debug files.

We suppose that the conservation law has the form $\omega = f_x dx + f_t dt$. Using the same `ansatz` as in the previous example we assume

```
fx:=(for each el in ansatz sum (c(ctel:=ctel+1)*el))$
ft:=(for each el in ansatz sum (c(ctel:=ctel+1)*el))$
```

Next we define the equation $\bar{d}(\omega) = 0$, where $\bar{d}$ is the total exterior derivative restricted to the equation.

```
equ 1:=ddt(fx)-ddx(ft)$
```

After solving the equation as in the above example we get

```
fx := c(3)*u_x1t0 + c(2)*u_x0t0 + c(1)$
ft := (2*c(8) + 2*c(3)*u_x0t0*u_x1t0 + 2*c(3)*u_x3t0 + c(2)*u_x0t0**2 +
2*c(2)*u_x2t0)/2$
```

Unfortunately it is clear that the conservation law corresponding to `c(3)` is trivial, because it is just the KdV equation. Here this fact is evident; how to get rid of less evident trivialities by an 'automatic' mechanism? We considered this problem in the file `kdv_lcl2`, where we solved the equation

```
equ 1:=fx-ddx(f0);
equ 2:=ft-ddt(f0);
```

after having loaded the values `fx` and `ft` found by the previous program. In order to do that we have to introduce two new counters:

```
operator cc,equ;
cctel:=0;
```

We make the following ansatz on `f0`:

```
f0:=(for each el in ansatz sum (cc(cctel:=cctel+1)*el))$
```

After solving the system, issuing the commands

```
fxnontriv := fx-ddx(f0);
ftnontriv := ft-ddt(f0);
```

we obtain

```
fxnontriv := c(2)*u_x0t0 + c(1)$
ftnontriv := (2*c(8) + c(2)*u_x0t0**2 + 2*c(2)*u_x2t0)/2$
```

This mechanism can be easily generalized to situations in which the conservation laws which are found by the program are difficult to treat by pen and paper. However, we will present another approach to the computation of conservation laws in subsection 7.2.

6. Local Hamiltonian operators

In this section we will show how to compute local Hamiltonian operators for Korteweg–de Vries, Boussinesq and Kadomtsev–Petviashvili equations. It is interesting to note that we will adopt the same computational scheme for both equations, even if the latter is not in evolutionary form and it has more than two independent variables. This comes from a new mathematical theory which started in [13] for evolution equations and was later extended to general differential equations in [14].

6.1. **Korteweg–de Vries equation.**

Here we will find local Hamiltonian operators for the KdV equation $u_t = u_{xxx} + uu_x$. Concretely, we have to solve $\bar{\ell}_{KdV}(\texttt{phi}) = 0$ over the equation

$$\begin{cases} u_t = u_{xxx} + uu_x \\ p_t = p_{xxx} + up_x \end{cases}$$

or, in geometric terminology, find the shadows of symmetries on the ℓ^*-covering of the KdV equation. The reference paper for this type of computations is [13].

The file containing this example is `kdv_lho1`.

We stress that the linearization $\bar{\ell}_{KdV}(\texttt{phi}) = 0$ is the equation

```
ddt(phi)-u_x0t0*ddx(phi)-u_x1t0*phi-ddx(ddx(ddx(phi)))=0
```

but the total derivatives are lifted to the ℓ^* covering, hence they must contain also derivatives with respect to p's. This will be achieved by treating p variables as odd and introducing the odd parts of `ddx` and `ddt`.

At this point we should discuss how CDIFF treats odd variables. Externally they look just like even variables, and are indicated by a letter followed by a multiindex. Internally, they are components of an operator: `ext(1)`, `ext(2)`, `ext(3)`, ..., and they are endowed with a skew-symmetric product. There are CDIFF commands which translate expressions involving odd variables. Namely, to replace in the expression `f` odd variables with `ext` variables (for example, for computations with CDIFF), do

```
replace_oddext(f);
```

and do

```
replace_extodd(g);
```

if you wish to translate a result g of CDIFF computations, depending on skew-symmetric internal variables `ext`, in a more readable form in terms of odd variables.

The ansatz must be generalized to odd variables.

```
graadlijst:=der_deg_ordering(0,all_parametric_der)$
graadlijst_odd:=der_deg_ordering(1,all_parametric_odd)$
graadmon:=for i:=1:10 collect mkvarlist1(i,i)$
graadmon:={1} . graadmon$
```

In particular, the unknown must be linear in odd variables, so we need a list of graded monomials which are linear in odd variables. The function `mkalllinodd` produces all monomials which are linear with respect to the variables from `graadlijst_odd`, have (monomial) coefficients from the variables in `graadlijst`, and have total scale degrees from 1 to 6. Such monomials are then converted to the internal representation of odd variables.

```
linodd:=mkalllinodd(graadmon,graadlijst_odd,1,6)$
linext:=replace_oddext(linodd)$
```

Note that all odd variables have positive scale degrees thanks to our initial choice `deg_odd_var:=1;`. Finally, the ansatz for local Hamiltonian operators:

```
sym:=(for each el in linext sum (c(ctel:=ctel+1)*el))$
```

After having set

```
equ 1:=ddt(sym)-u_x0t0*ddx(sym)-u_x1t0*sym-ddx(ddx(ddx(sym)));
```

and having initialized the equation solver as before, we do `splitext`

```
splitext 1;
```

in order to split the polynomial equation with respect to the `ext` variables, then `splitvars`

```
tel1:=tel;
for i:=2:tel1 do begin splitvars i;equ i:=0;end;
```

in order to split the resulting polynomial equation in a list of equations on the coefficients of all monomials.

Now we are ready to solve all equations:

```
pte tel;
for i:=2:tel do es i;
end;
```

Note that we want *all* equations to be solved!

The results are the two well-known Hamiltonian operators for the KdV:

```
sym := (3*c(5)*ext(4) + 2*c(5)*ext(2)*u_x0t0 + c(5)*ext(1)*u_x1t0 +
3*c(2)*ext(2))/3$
sym_odd := (c(5)*p_x0t0*u_x1t0 + 2*c(5)*p_x1t0*u_x0t0 + 3*c(5)*p_x3t0 +
3*c(2)*p_x1t0)/3$
```

Note the internal and external expressions of the result. Of course, the results correspond to the operators

$$\texttt{ext(4)} \to D_x,$$

$$\texttt{3*c(3)*ext(6) + 2*c(3)*ext(4)*u + c(3)*ext(3)*u1} \to 3D_{xxx} + 2uD_x + u_x$$

Note that each operator is multiplied by one arbitrary real constant, `c(5)` and `c(2)`.

6.2. **Boussinesq equation.** There is no conceptual difference when computing for systems of PDEs with respect to the previous computations for scalar equations. We will look for Hamiltonian structures for the following Boussinesq equation:

$$\begin{cases} u_t - u_x v - u v_x - \sigma v_{xxx} = 0 \\ v_t - u_x - v v_x = 0 \end{cases} \tag{2}$$

where σ is a constant. This example also shows how to deal with jet spaces with more than one dependent variable. Here gradings can be taken as

$$[t] = -2, \quad [x] = -1, \quad [v] = 1, \quad [u] = 2, \quad [p] = [\frac{\partial}{\partial u}] = -2, \quad [q] = [\frac{\partial}{\partial v}] = -1$$

where p, q are the two coordinates in the space of generating functions of conservation laws.

The linearization of the above system and its adjoint are, respectively

$$\ell_{\mathrm{Bou}} = \begin{pmatrix} D_t - vD_x - v_x & -u_x - uD_x - \sigma D_{xxx} \\ -D_x & D_t - v_x - vD_x \end{pmatrix}, \ \ell^*_{\mathrm{Bou}} = \begin{pmatrix} -D_t + vD_x & D_x \\ uD_x + \sigma D_{xxx} & -D_t + vD_x \end{pmatrix}$$

and lead to the ℓ^*_{Bou} covering equation

$$\begin{cases} -p_t + vp_x + q_x = 0 \\ up_x + \sigma p_{xxx} - q_t + vq_x = 0 \\ u_t - u_x v - uv_x - \sigma v_{xxx} = 0 \\ v_t - u_x - vv_x = 0 \end{cases}$$

We have to find shadows of symmetries on the above covering. At the level of source file (`bou_lho1_test`) the input data is:

```
indep_var:={x,t}$
dep_var:={u,v}$
odd_var:={p,q}$
deg_indep_var:={-1,-2}$
deg_dep_var:={2,1}$
deg_odd_var:={1,2}$
```

```
total_order:=8$
principal_der:={u_x0t1,v_x0t1}$
de:={u_x1t0*v_x0t0+u_x0t0*v_x1t0+sig*v_x3t0,u_x1t0+v_x0t0*v_x1t0}$
principal_odd:={p_x0t1,q_x0t1}$
de_odd:={v_x0t0*p_x1t0+q_x1t0,u_x0t0*p_x1t0+sig*p_x3t0+v_x0t0*q_x1t0}$
```

The ansatz for the components of the Hamiltonian operator, of scale degree between 1 and 6, is

```
linodd:=mkalllinodd(graadmon,graadlijst_odd,1,6)$
linext:=replace_oddext(linodd)$
phi1:=(for each el in linext sum (c(ctel:=ctel+1)*el))$
phi2:=(for each el in linext sum (c(ctel:=ctel+1)*el))$
```

and the equation for shadows of symmetries is

```
equ 1:=ddt(phi1)-v_x0t0*ddx(phi1)-v_x1t0*phi1-u_x1t0*phi2-
u_x0t0*ddx(phi2)-sig*ddx(ddx(ddx(phi2)));
equ 2:=-ddx(phi1)-v_x0t0*ddx(phi2)-v_x1t0*phi2+ddt(phi2);
```

After the usual procedures for decomposing polynomials we obtain three local Hamiltonian operators:

```
phi1_odd := (2*c(27)*p_x0t0*sig*v_x3t0 + 2*c(27)*p_x0t0*u_x0t0*v_x1t0 +
2*c(27)*p_x0t0*u_x1t0*v_x0t0 + 6*c(27)*p_x1t0*sig*v_x2t0 +
4*c(27)*p_x1t0*u_x0t0*v_x0t0 + 6*c(27)*p_x2t0*sig*v_x1t0 +
4*c(27)*p_x3t0*sig*v_x0t0 + 2*c(27)*q_x0t0*u_x1t0 +
4*c(27)*q_x1t0*u_x0t0 + c(27)*q_x1t0*v_x0t0**2 + 4*c(27)*q_x3t0*sig +
2*c(13)*p_x0t0*u_x1t0 + 4*c(13)*p_x1t0*u_x0t0 + 4*c(13)*p_x3t0*sig +
2*c(13)*q_x1t0*v_x0t0 + 4*c(5)*q_x1t0*sig)/(4*sig)$

phi2_odd := (2*c(27)*p_x0t0*u_x1t0 + 2*c(27)*p_x0t0*v_x0t0*v_x1t0 +
4*c(27)*p_x1t0*u_x0t0 + c(27)*p_x1t0*v_x0t0**2 + 4*c(27)*p_x3t0*sig +
2*c(27)*q_x0t0*v_x1t0 + 4*c(27)*q_x1t0*v_x0t0 + 2*c(13)*p_x0t0*v_x1t0 +
2*c(13)*p_x1t0*v_x0t0 + 4*c(13)*q_x1t0 + 4*c(5)*p_x1t0*sig)/(4*sig)$
```

There is a whole hierarchy of nonlocal Hamiltonian operators [13].

6.3. **Kadomtsev–Petviashvili equation.** There is no conceptual difference in symbolic computations of Hamiltonian operators for PDEs in 2 independent variables and in more than 2 independent variables, regardless of the fact that the equation at hand is written in evolutionary form. As a model example, we consider the KP equation

$$u_{yy} = u_{tx} - u_x^2 - uu_{xx} - \frac{1}{12}u_{xxxx}. \tag{3}$$

Proceeding as in the above examples we input the following data:

```
indep_var:={t,x,y}$
dep_var:={u}$
odd_var:={p}$
deg_indep_var:={-3,-2,-1}$
deg_dep_var:={2}$
deg_odd_var:={1}$
```

```
total_order:=6$
principal_der:={u_t0x0y2}$
de:={u_t1x1y0-u_t0x1y0**2-u_t0x0y0*u_t0x2y0-(1/12)*u_t0x4y0}$
principal_odd:={p_t0x0y2}$
de_odd:={p_t1x1y0-u_t0x0y0*p_t0x2y0-(1/12)*p_t0x4y0}$
```

and look for Hamiltonian operators of scale degree between 1 and 5:

```
linodd:=mkalllinodd(graadmon,graadlijst_odd,1,5)$
linext:=replace_oddext(linodd)$
phi:=(for each el in linext sum (c(ctel:=ctel+1)*el))$
```

After solving the equation for shadows of symmetries in the cotangent covering

```
equ 1:=ddy(ddy(phi))-ddt(ddx(phi))+2*u_t0x1y0*ddx(phi)
+u_t0x2y0*phi+u_t0x0y0*ddx(ddx(phi))
+(1/12)*ddx(ddx(ddx(ddx(phi))))$
```

we get the only local Hamiltonian operator

```
phi_odd := c(10)*p_t0x2y0$
```

As far as we know there are no further local Hamiltonian operators.

Remark: the above Hamiltonian operator is already known in an evolutionary presentation of the KP equation [17]. Our mathematical theory of Hamiltonian operators for general differential equations [14] allows us to formulate and solve the problem for any presentation of the KP equation. Change of coordinate formulae could also be provided.

7. Non-local Hamiltonian operators

In this section we will show an experimental way to find nonlocal Hamiltonian operators. The word 'experimental' comes from the lack of a comprehensive mathematical theory of nonlocal Hamiltonian operators. In any case we will achieve the results by means of a covering of the cotangent covering. Indeed, it can be proved that there is a $1-1$ correspondence between (higher) symmetries of the initial equation and conservation laws on the cotangent covering. Such conservation laws provide new potential variables, hence a covering (see [12] for theoretical details on coverings).

In Section 7.2 we will also discuss a procedure for finding conservation laws from their generating functions that is of independent interest.

7.1. Korteweg–de Vries equation.

Here we will compute some nonlocal Hamiltonian operators for the KdV equation. The result of the computation (without the details below) has been published in [13].

We have to solve equations of the type `ddx(ct)-ddt(cx)` as in 5. The main difference is that we will attempt a solution on the ℓ^*-covering (see Subsection 6). For this reason, first of all we have to determine covering variables with the usual mechanism of introducing them through conservation laws, this time on the ℓ^*-covering.

As a first step, let us compute conservation laws on the ℓ^*-covering whose components are linear in the p's. This computation can be found in the file `kdv_nlcl1` and related results and debug files.

The conservation laws that we are looking for are in $1-1$ correspondence with symmetries of the initial equation [13]. We will look for conservatoin laws which correspond to Galilean boost, x-translation, t-translation at the same time. In the case of 2 independent variables and 1 dependent variable, one could prove that one component of such conservation laws can always be written as `sym*p_x0t0` as follows:

```
c1x_odd:=(t*u_x1t0+1)*p_x0t0$ % degree 1
c2x_odd:=u_x1t0*p_x0t0$ % degree 4
c3x_odd:=(u_x0t0*u_x1t0+u_x3t0)*p_x0t0$ % degree 6
```

Of course, we must pass to the internal representation:

```
c1x:=replace_oddext(c1x_odd)$
c2x:=replace_oddext(c2x_odd)$
c3x:=replace_oddext(c3x_odd)$
```

The second component must be found by solving an equation. To this aim we produce the ansatz

```
c1t_odd:=f1*p_x0t0+f2*p_x1t0+f3*p_x2t0$
c1t:=replace_oddext(c1t_odd)$
c2t:=(for each el in linext6 sum (c(ctel:=ctel+1)*el))$ % degree 6
c3t:=(for each el in linext8 sum (c(ctel:=ctel+1)*el))$ % degree 8
```

where we already introduced the sets `linext6` and `linext8` of 6-th and 8-th degree monomials which are linear in odd variables (see the source code). For the first conservation law solutions of the equation

```
equ 1:=ddx(c1t)-ddt(c1x);
```

are found by hand due to the presence of 't' in the symmetry:

```
f3:=t*u_x1t0+1$
f2:=-ddx(f3)$
f1:=u_x0t0*f3+ddx(ddx(f3))$
```

We also have the equations

```
equ 2:=ddx(c2t)-ddt(c2x);
equ 3:=ddx(c3t)-ddt(c3x);
```

They are solved in the usual way (see the source code of the example and the results file `kdv_nlcl1_res`).

Now, we solve the equation for shadows of nonlocal symmetries in a covering of the ℓ^*-covering (source file `kdv_nlho1_test`). We can produce such a covering by introducing three new nonlocal (potential) variables `ra,rb,rc`. We are going to look for non-local Hamiltonian operators depending linearly on one of these variables. To this aim we modify the odd part of the equation to include the components of the above conservation laws as the derivatives of the new non-local variables `ra`, `rb`, `rc`:

```
principal_odd:={p_x0t1,r1_x1t0,r1_x0t1,r2_x1t0,r2_x0t1,r3_x1t0,r3_x0t1}$
de_odd:={u_x0t0*p_x1t0+p_x3t0,
(t*u_x1t0+1)*p_x0t0,
p_x2t0*t*u_x1t0 + p_x2t0 - p_x1t0*t*u_x2t0
+ p_x0t0*t*u_x0t0*u_x1t0 + p_x0t0*t*u_x3t0 + p_x0t0*u_x0t0,
u_x1t0*p_x0t0,
```

```
p_x2t0*u_x1t0 - p_x1t0*u_x2t0 + p_x0t0*u_x0t0*u_x1t0
 + p_x0t0*u_x3t0,
(u_x0t0*u_x1t0+u_x3t0)*p_x0t0,
p_x2t0*u_x0t0*u_x1t0 + p_x2t0*u_x3t0 - p_x1t0*u_x0t0*u_x2t0
- p_x1t0*u_x1t0**2 - p_x1t0*u_x4t0 + p_x0t0*u_x0t0**2*u_x1t0
+ 2*p_x0t0*u_x0t0*u_x3t0 + 3*p_x0t0*u_x1t0*u_x2t0 + p_x0t0*u_x5t0}$
```

The scale degree analysis of the local Hamiltonian operators of the KdV equation leads to the formulation of the ansatz

```
phi:=(for each el in linext sum (c(ctel:=ctel+1)*el))$
```

where `linext` is the list of graded mononials which are linear in odd variables and have degree 7 (see the source file). The equation for shadows of nonlocal symmetries in ℓ^*-covering

```
equ 1:=ddt(phi)-u_x0t0*ddx(phi)-u_x1t0*phi-ddx(ddx(ddx(phi)));
```

is solved in the usual way, obtaining (in odd variables notation):

```
phi_odd := (c(1)*(4*p_x0t0*u_x0t0*u_x1t0 + 3*p_x0t0*u_x3t0 +
4*p_x1t0*u_x0t0**2 + 12*p_x1t0*u_x2t0 + 18*p_x2t0*u_x1t0 +
12*p_x3t0*u_x0t0 + 9*p_x5t0 - r2_x0t0*u_x1t0))/9$
```

Higher non-local Hamiltonian operators could also be found [13].

7.2. **Plebanski equation.** The Plebanski (or second Heavenly) equation

$$F = u_{tt}u_{xx} - u_{tx}^2 + u_{xz} + u_{ty} = 0 \qquad (4)$$

is Lagrangian, hence it admits a trivial local Hamiltonian operator which is just the Noether map. Nonlocal Hamiltonian operators have been computed in an evolutionary presentation of the equation in [19]. We can recompute such operators in the above Lagrangian presentation by introducing a suitable nonlocal variable on the cotangent covering. Namely, we compute a linear conservation law (with respect to p's) on the cotangent covering which corresponds with the u-translation symmetry (see [16] for a theoretical description). After guessing the generating function of the conservation law $\psi = (0, 1)$ from the generating function $\varphi = 1$ of the u-translation symmetry , we deduce that the equation

$$\bar{d}\omega = \ell_F^*(p) \qquad (5)$$

should hold on the jet space. Here

$$\omega = ct\, dx \wedge dy \wedge dz + cx\, dt \wedge dy \wedge dz + cy\, dt \wedge dx \wedge dz + cz\, dt \wedge dx \wedge dy,$$

where ct, cx, cy, cz are linear functions of p's and its derivatives, with coefficients in u's[2], and

$$\bar{d}\omega = (D_t ct - D_x cx + D_y cy - D_z cz) dt \wedge dx \wedge dy \wedge dz,$$

where total derivatives are lifted on the jet space of even and odd coordinates.

Then, we try to find representatives of the above conservation law which have not more than two non-vanishing components. In particular we will solve the equation

$$D_t ct - D_x cx = 0 \qquad (6)$$

[2]In general, coefficients can explicitly depend on independent variables.

in the cotangent covering. Such an equation cannot be solved in general, but it can be solved in this case. In order to solve it we perform dimensional analysis on (5) and we deduce the gradings of ct and cx. The result is determined up to trivial conservation laws, so that we have to remove them; at the end we remain with a single 2-component conservation law.

This allows us to introduce a new nonlocal *odd* variable r on the cotangent covering such that $r_x = ct$, $r_t = cx$. We obtain an Abelian covering of the cotangent covering:

$$\begin{cases} r_x = ct, \\ r_t = cx, \\ \ell_F^*(p) = 0, \\ F = 0. \end{cases}$$

A nonlocal Hamiltonian operator will be a shadow of symmetry of the above system with respect to the initial equation $F = 0$ with the property of being linear with respect to all (p's and r's) odd variables. With the above nonlocal variable we find a nonlocal Hamiltonian operator which, after changing coordinates to the evolutionary presentation of [19], coincides with one of the nonlocal Hamiltonian operators presented in that paper[3].

Let us describe the computation in detail. We start with the conservation law (see the file `ple_nlcl1.red`):

```
indep_var:={t,x,y,z}$
dep_var:={u}$
odd_var:={p}$
deg_indep_var:={-1,-1,-4,-4}$
deg_dep_var:={1}$
deg_odd_var:={4}$
total_order:=6$

% left-hand side of the differential equation
principal_der:={u_t0x1y0z1}$
% right-hand side of the differential equation
de:={-u_t1x0y1z0+u_t1x1y0z0**2-u_t2x0y0z0*u_t0x2y0z0}$
% same constructions for odd coordinates
principal_odd:={p_t0x1y0z1}$
de_odd:={-p_t1x0y1z0+2*u_t1x1y0z0*p_t1x1y0z0-u_t0x2y0z0*p_t2x0y0z0-
u_t2x0y0z0*p_t0x2y0z0}$
```

Now we limit the computation to variables of jetspace order not greater than 4; this is done through the function `selectvars`, which takes four arguments. The first can be 0 for even variables or 1 for odd variables, the second argument is the specified order, the third argument is the subset of dependent variables that we wish to select (all dependent variables in our case) and the fourth is the set of derivative coordinates from which we wish to extract the variables.

[3]We observe that in [19] also the trivial Hamiltonian operator is recovered in the evolutionary presentation; of course it has an apparently nontrivial expression.

```
v0_4:=for i:=0:4 join selectvars(0,i,dep_var,all_parametric_der)$
vo0_4:=for i:=0:4 join selectvars(1,i,odd_var,all_parametric_odd)$
```

We rearrange all variables by their scale degree, starting from variables of degree 1 (the degree is always chosen in such a way that grading of any even or odd derivative coordinates is positive):

```
graadlijst:=der_deg_ordering(0,v0_4)$
graadlijst_odd:=der_deg_ordering(1,vo0_4)$
```

and we collect graded monomials of scale degree less than or equal 13:

```
graadmon:=for i:=1:13 collect mkvarlist1(i,i)$
graadmon:={1} . graadmon$
```

Then we have to make an ansatz for the conservation law: since the summands of `ellstarfp` have degree 9 we assume `[ct]=[cx]=8`

```
deg_cx:=8$
deg_ct:=deg_cx$
```

It would also be `[cy]=[cz]=5`, but in this computation we assume $cy = cz = 0$. Note that no simplification can be assumed like in the case of 2 independent variables: it is not true, in general, that one component of such conservation laws can always be written as `sym*p_x0t0`. The ansatz is constructed through the function `mklinodd`, which takes three arguments: the list of lists of graded monomials of degree 1, 2, ..., the list of lists of graded odd variables of degree 1, 2, ..., and the final degree of their products:

```
linoddt:=mklinodd(graadmon,graadlijst_odd,deg_ct)$
linoddx:=linoddt$
linextt:=replace_oddext(linoddt)$
linextx:=linextt$
% Ansatz:
ct:=(for each el in linextt sum (c(ctel:=ctel+1)*el))$
cx:=(for each el in linextx sum (c(ctel:=ctel+1)*el))$
```

The equation for conservation laws can be checked for the presence of `letop`. If an error is issued, the computation must be rerun with a higher value of `total_order`:

```
ct_t:=ddt(ct)$
cx_x:=ddx(cx)$
check_letop({ct_t,cx_x})$
```

Note that in the folder containing all examples there is also a shell script, `rrr.sh` (works only under `bash`, a GNU/Linux command interpreter) which can be used to run reduce on a given CDIFF program. If the function `check_letop` issues an error message then the script reruns the computation with a new value of `total_order` one unity higher than the previous one.

Finally we define the equation

```
equ 1:=ct_t-cx_x$
```

The equation admits a lot of solutions, almost all of which are trivial conservation laws (here they are expressed in odd variables):

```
ct_odd := c(32)*p_t0x0y0z0*u_t2x1y0z0 + c(32)*p_t0x1y0z0*u_t2x0y0z0 + c(30)*
p_t0x0y0z0*u_t1x2y0z0 + c(30)*p_t0x1y0z0*u_t1x1y0z0 + c(28)*p_t0x0y0z0*
```

```
u_t0x3y0z0 + c(28)*p_t0x1y0z0*u_t0x2y0z0 + c(26)*p_t0x0y0z0*u_t0x0y0z0*
u_t1x1y0z0 + c(26)*p_t0x0y0z0*u_t0x1y0z0*u_t1x0y0z0 + c(26)*p_t0x1y0z0*
u_t0x0y0z0*u_t1x0y0z0 + c(24)*p_t0x0y0z0*u_t0x0y0z0*u_t0x2y0z0 + c(24)*
p_t0x0y0z0*u_t0x1y0z0**2 + c(24)*p_t0x1y0z0*u_t0x0y0z0*u_t0x1y0z0 + 3*c(22)*
p_t0x0y0z0*u_t0x0y0z0**2*u_t0x1y0z0 + c(22)*p_t0x1y0z0*u_t0x0y0z0**3 - c(20)*
p_t0x0y0z0*u_t1x2y0z0 + c(20)*p_t0x2y0z0*u_t1x0y0z0 + c(19)*p_t1x0y0z0*
u_t1x1y0z0 + c(19)*p_t1x1y0z0*u_t1x0y0z0 - c(17)*p_t0x0y0z0*u_t0x3y0z0 + c(17)*
p_t0x2y0z0*u_t0x1y0z0 + c(16)*p_t1x0y0z0*u_t0x2y0z0 + c(16)*p_t1x1y0z0*
u_t0x1y0z0 - 2*c(14)*p_t0x0y0z0*u_t0x0y0z0*u_t0x2y0z0 - 2*c(14)*p_t0x0y0z0*
u_t0x1y0z0**2 + c(14)*p_t0x2y0z0*u_t0x0y0z0**2 + 2*c(13)*p_t1x0y0z0*u_t0x0y0z0*
u_t0x1y0z0 + c(13)*p_t1x1y0z0*u_t0x0y0z0**2 + c(11)*p_t0x0y0z0*u_t0x3y0z0 + c(11
)*p_t0x3y0z0*u_t0x0y0z0 - c(10)*p_t1x0y0z0*u_t0x2y0z0 + c(10)*p_t1x2y0z0*
u_t0x0y0z0 + c(9)*p_t2x0y0z0*u_t0x1y0z0 + c(9)*p_t2x1y0z0*u_t0x0y0z0 + c(7)*
p_t0x4y0z0 + c(6)*p_t1x3y0z0 + c(5)*p_t2x2y0z0 + c(4)*p_t3x1y0z0 + c(1)*
p_t0x0y0z0*u_t1x2y0z0 + c(1)*p_t0x0y1z0 + c(1)*p_t1x0y0z0*u_t0x2y0z0$

cx_odd := c(32)*p_t0x0y0z0*u_t3x0y0z0 + c(32)*p_t1x0y0z0*u_t2x0y0z0 + c(30)*
p_t0x0y0z0*u_t2x1y0z0 + c(30)*p_t1x0y0z0*u_t1x1y0z0 + c(28)*p_t0x0y0z0*
u_t1x2y0z0 + c(28)*p_t1x0y0z0*u_t0x2y0z0 + c(26)*p_t0x0y0z0*u_t0x0y0z0*
u_t2x0y0z0 + c(26)*p_t0x0y0z0*u_t1x0y0z0**2 + c(26)*p_t1x0y0z0*u_t0x0y0z0*
u_t1x0y0z0 + c(24)*p_t0x0y0z0*u_t0x0y0z0*u_t1x1y0z0 + c(24)*p_t0x0y0z0*
u_t0x1y0z0*u_t1x0y0z0 + c(24)*p_t1x0y0z0*u_t0x0y0z0*u_t0x1y0z0 + 3*c(22)*
p_t0x0y0z0*u_t0x0y0z0**2*u_t1x0y0z0 + c(22)*p_t1x0y0z0*u_t0x0y0z0**3 - c(20)*
p_t0x0y0z0*u_t2x1y0z0 + c(20)*p_t0x1y0z0*u_t2x0y0z0 - c(20)*p_t1x0y0z0*
u_t1x1y0z0 + c(20)*p_t1x1y0z0*u_t1x0y0z0 + c(19)*p_t1x0y0z0*u_t2x0y0z0 + c(19)*
p_t2x0y0z0*u_t1x0y0z0 - c(17)*p_t0x0y0z0*u_t1x2y0z0 + c(17)*p_t0x1y0z0*
u_t1x1y0z0 - c(17)*p_t1x0y0z0*u_t0x2y0z0 + c(17)*p_t1x1y0z0*u_t0x1y0z0 + c(16)*
p_t1x0y0z0*u_t1x1y0z0 + c(16)*p_t2x0y0z0*u_t0x1y0z0 - 2*c(14)*p_t0x0y0z0*
u_t0x0y0z0*u_t1x1y0z0 - 2*c(14)*p_t0x0y0z0*u_t0x1y0z0*u_t1x0y0z0 + 2*c(14)*
p_t0x1y0z0*u_t0x0y0z0*u_t1x0y0z0 - 2*c(14)*p_t1x0y0z0*u_t0x0y0z0*u_t0x1y0z0 + c(
14)*p_t1x1y0z0*u_t0x0y0z0**2 + 2*c(13)*p_t1x0y0z0*u_t0x0y0z0*u_t1x0y0z0 + c(13)*
p_t2x0y0z0*u_t0x0y0z0**2 + c(11)*p_t0x0y0z0*u_t1x2y0z0 - c(11)*p_t0x1y0z0*
u_t1x1y0z0 + c(11)*p_t0x2y0z0*u_t1x0y0z0 + c(11)*p_t1x0y0z0*u_t0x2y0z0 - c(11)*
p_t1x1y0z0*u_t0x1y0z0 + c(11)*p_t1x2y0z0*u_t0x0y0z0 - c(10)*p_t1x0y0z0*
u_t1x1y0z0 + c(10)*p_t1x1y0z0*u_t1x0y0z0 - c(10)*p_t2x0y0z0*u_t0x1y0z0 + c(10)*
p_t2x1y0z0*u_t0x0y0z0 + c(9)*p_t2x0y0z0*u_t1x0y0z0 + c(9)*p_t3x0y0z0*u_t0x0y0z0
+ c(7)*p_t1x3y0z0 + c(6)*p_t2x2y0z0 + c(5)*p_t3x1y0z0 + c(4)*p_t4x0y0z0 + c(1)*
p_t0x0y0z0*u_t2x1y0z0 - c(1)*p_t0x0y0z1 - c(1)*p_t0x1y0z0*u_t2x0y0z0 + 2*c(1)*
p_t1x0y0z0*u_t1x1y0z0$
```

We begin the removal of trivial conservation laws from the above solution. The idea is that a conservation law (*i.e.* a horizontal 3-form) is trivial if it is equal to the horizontal differential of a 2-form. Such two form will be chosen according with dimensional analysis. We introduce an operator and a counter that will parametrize trivial conservation laws:

```
operator cc$
```

```
cctel:=0$
```

Then we assume that the trivial conservation law has the form

$$tcl = tctxdt \wedge dx + tctydt \wedge dy + tctzdt \wedge dz + tcxydx \wedge dy + tcxzdx \wedge dz + tcyzdy \wedge dz$$

so that a conservation law will be trivial if and only if

$$\begin{aligned}\bar{d}(tcl) =&(D_z(tcxy) - D_y(tcxz) + D_x(tcyz))dx \wedge dy \wedge dz+\\ &(D_z(tcty) - D_y(tctz) + D_t(tcyz))dt \wedge dy \wedge dz+\\ &(D_z(tctx) - D_x(tctz) + D_t(tcxz))dt \wedge dx \wedge dz+\\ &(D_y(tctx) - D_x(tcty) + D_t(tcxy))dt \wedge dx \wedge dy\\ =&ctdx \wedge dy \wedge dz + cxdt \wedge dy \wedge dz + cydt \wedge dx \wedge dz + czdt \wedge dx \wedge dy\end{aligned}$$

Since in our case we are looking for a 2-component conservation law, we will assume a single potential of the form: $tcyzdy \wedge dz$:

```
deg_tcyz:=7$
linodd_tcyz:=mklinodd(graadmon,graadlijst_odd,deg_tcyz)$
linext_tcyz:=replace_oddext(linodd_tcyz)$
tcyz:=(for each el in linext_tcyz sum (cc(cctel:=cctel+1)*el))$
```

After clearing the previous equations, we set up the new equation

```
clear equ$
operator equ$

equ 1:=ddx(tcyz) - ct$
equ 2:=ddt(tcyz) - cx$
```

Note that in this case if the equation can be solved then the conservation law is trivial; only if the equation *cannot* be solved we found at least one nontrivial conservation law. Results can be written as follows:

```
write ctnontriv:=equ 1$
write cxnontriv:=equ 2$
```

they will be nonzero if a nontrivial conservation law remains in ct and cx.

Now, we look for nonlocal Hamiltonian operators in the cotangent covering using a new nonlocal odd variable r as follows (see `ple_nlho1.red`):

```
indep_var:={t,x,y,z}$
dep_var:={u}$
odd_var:={p,r}$
deg_indep_var:={-1,-1,-4,-4}$
deg_dep_var:={1}$
deg_odd_var:={1,4}$
total_order:=6$
principal_der:={u_t0x1y0z1}$
de:={-u_t1x0y1z0+u_t1x1y0z0**2-u_t2x0y0z0*u_t0x2y0z0}$
% rhs of the equations that define the nonlocal variable
r_t:=p_t0x0y0z1 + p_t0x1y0z0*u_t2x0y0z0 + p_t2x0y0z0*u_t0x1y0z0$
r_x:=- p_t0x0y1z0 + p_t0x1y0z0*u_t1x1y0z0 + p_t1x1y0z0*u_t0x1y0z0$
```

```
% We add conservation laws as new nonlocal odd variables;
principal_odd:={
p_t0x1y0z1,
r_t0x1y0z0,r_t1x0y0z0
}$
%
de_odd:={-p_t1x0y1z0+2*u_t1x1y0z0*p_t1x1y0z0-u_t0x2y0z0*p_t2x0y0z0-
u_t2x0y0z0*p_t0x2y0z0,
r_x,r_t
}$
```

We look for Hamiltonian operators which depend on r (which has scale degree 4); we produce the following ansatz for `phi`:

```
linodd:=mkalllinodd_e(graadmon,graadlijst_odd,1,4)$
linext:=replace_oddext_e(linodd)$
phi:=(for each el in linext sum (c(ctel:=ctel+1)*el))$
```

then we solve the equation of shadows of symmetries:

```
equ 1:=ddx(ddz(phi))+ddt(ddy(phi))-2*u_t1x1y0z0*ddt(ddx(phi))
+u_t0x2y0z0*ddt(ddt(phi))+u_t2x0y0z0*ddx(ddx(phi))$
```

The solution in odd coordinates is

```
phi_odd :=  - c(13)*p_t1x0y0z0*u_t0x1y0z0 + c(13)*r_t0x0y0z0
+ c(1)*p_t0x0y0z0$
```

hence we obtain the Noether map (the identity operator p) and the new nonlocal operator $r - u_x p_t$. It can be proved that changing coordinates to the evolutionary presentation yields the local operator (which has a much more complex expression than the identity operator) and one of the nonlocal operators of [19].

References

[1] Obtaining REDUCE: http://reduce-algebra.sourceforge.net/.

[2] Geometry of Differential Equations web site: http://gdeq.org.

[3] GetDeb website: http://archive.getdeb.net/getdeb/ubuntu/pool/apps/r/reduce-algebra/.

[4] notepad++: http://notepad-plus.sourceforge.net/

[5] List of text editors: http://en.wikipedia.org/wiki/List_of_text_editors

[6] How to install emacs in Windows: http://www.cmc.edu/math/alee/emacs/emacs.html. See also http://www.gnu.org/software/emacs/windows/ntemacs.html

[7] How to install REDUCE in Windows: http://reduce-algebra.sourceforge.net/windows.html

[8] G.H.M. Roelofs, The SUPER_VECTORFIELD package for REDUCE. Version 1.0, Memorandum 1099, Dept. Appl. Math., University of Twente, 1992. Available at http://gdeq.org.

[9] G.H.M. Roelofs, The INTEGRATOR package for REDUCE. Version 1.0, Memorandum 1100, Dept. Appl. Math., University of Twente, 1992. Available at http://gdeq.org.

[10] G.F. Post, A manual for the package TOOLS 2.1, Memorandum 1331, Dept. Appl. Math., University of Twente, 1996. Available at http://gdeq.org.

[11] REDUCE IDE for emacs: http://centaur.maths.qmul.ac.uk/Emacs/REDUCE_IDE/

[12] A. V. Bocharov, V. N. Chetverikov, S. V. Duzhin, N. G. Khor′kova, I. S. Krasil′shchik, A. V. Samokhin, Yu. N. Torkhov, A. M. Verbovetsky and A. M. Vinogradov: Symmetries and Conservation Laws for Differential Equations of Mathematical Physics, I. S. Krasil′shchik and A. M. Vinogradov eds., Translations of Math. Monographs **182**, Amer. Math. Soc. (1999).

[13] P.H.M. Kersten, I.S. Krasil'shchik, A.M. Verbovetsky, *Hamiltonian operators and ℓ^*-covering*, Journal of Geometry and Physics **50** (2004), 273–302.

[14] P. Kersten, I. Krasil′shchik, A. Verbovetsky, and R. Vitolo, *Hamiltonian structures for general PDEs*, Differential equations: Geometry, Symmetries and Integrability. The Abel Symposium 2008 (B. Kruglikov, V. V. Lychagin, and E. Straume, eds.), Springer-Verlag, 2009, pp. 187–198, arXiv:0812.4895.

[15] I. Krasil′shchik and A. Verbovetsky, *Geometry of jet spaces and integrable systems*, J. Geom. Phys. (2011) doi:10.1016/j.geomphys.2010.10.012, arXiv:1002.0077.

[16] I. Krasil′shchik, A. Verbovetsky, R. Vitolo, *A unified approach to computation of integrable structures*, submitted to Acta Appl. Math. (2012).

[17] *Geometric Hamiltonian forms for the Kadomtsev–Petviashvili and Zabolotskaya–Khokhlov equations*, in Geometry in Partial Differential Equations, A. Prastaro, Th.M. Rassias eds., World Scientific (1994), 155–172.

[18] M. Marvan, *Sufficient set of integrability conditions of an orthonomic system*. Foundations of Computational Mathematics **9** (2009), 651–674.

[19] F. Neyzi, Y. Nutku, and M.B. Sheftel, *Multi-Hamiltonian structure of Plebanski's second heavenly equation* J. Phys. A: Math. Gen. **38** (2005), 8473. arXiv:nlin/0505030v2.

R. Vitolo, Dept. of Mathematics "E. De Giorgi", Università del Salento, via per Arnesano, 73100 Lecce, Italy

E-mail address: raffaele.vitolo@unisalento.it